```
(* The somewhat arbitrary velocity field we want as the solution *)

u = FullSimplify[{1 - A Cos[π (x - t)] Sin[π (y - t)] Exp[-2 ν π^2 t],
    1 + A Sin[π (x - t)] Cos[π (y - t)] Exp[-2 π^2 ν t]}];
MatrixForm[
 u]
```

$$\begin{pmatrix} 1 - A\, e^{-2 \pi^2 t \nu} \mathrm{Cos}[\pi\,(-t + x)]\, \mathrm{Sin}[\pi\,(-t + y)] \\ 1 + A\, e^{-2 \pi^2 t \nu} \mathrm{Cos}[\pi\,(-t + y)]\, \mathrm{Sin}[\pi\,(-t + x)] \end{pmatrix}$$

```
(* I think it looks fairly interesting *)

VectorPlot[u /. {A → 1, ν → 1 / 8, t → 0},
 {x, -1, 1}, {y, -1, 1}, VectorColorFunction → Hue]
```




```
(* Check the divergence of the velocity field *)

divu = FullSimplify[D[u[[1]], x] + D[u[[2]], y]]
```

0

```
(* Constructing terms in the Navier-Stokes equation,
starting from the time derivative of the velocity *)

dudt = {D[u[[1]], t], D[u[[2]], t]};
MatrixForm[dudt]
```

$$\begin{pmatrix} A\, e^{-2 \pi^2 t \nu} \pi\, \mathrm{Cos}[\pi\,(-t + x)]\, \mathrm{Cos}[\pi\,(-t + y)] + 2 A\, e^{-2 \pi^2 t \nu} \pi^2 \nu\, \mathrm{Cos}[\pi\,(-t + x)]\, \mathrm{Sin}[\pi\,(-t + y)] - A\, e^{-2} \\ -A\, e^{-2 \pi^2 t \nu} \pi\, \mathrm{Cos}[\pi\,(-t + x)]\, \mathrm{Cos}[\pi\,(-t + y)] - 2 A\, e^{-2 \pi^2 t \nu} \pi^2 \nu\, \mathrm{Cos}[\pi\,(-t + y)]\, \mathrm{Sin}[\pi\,(-t + x)] + A\, e^{-2} \end{pmatrix}$$

```
(* Gradient of the velocity field *)

gradu = FullSimplify[{{D[u[[1]], x], D[u[[1]], y]}, {D[u[[2]], x], D[u[[2]], y]}}];
MatrixForm[gradu]
```

$$\begin{pmatrix} A\, e^{-2 \pi^2 t \nu} \pi\, \mathrm{Sin}[\pi\,(-t + x)]\, \mathrm{Sin}[\pi\,(-t + y)] & -A\, e^{-2 \pi^2 t \nu} \pi\, \mathrm{Cos}[\pi\,(-t + x)]\, \mathrm{Cos}[\pi\,(-t + y)] \\ A\, e^{-2 \pi^2 t \nu} \pi\, \mathrm{Cos}[\pi\,(-t + x)]\, \mathrm{Cos}[\pi\,(-t + y)] & -A\, e^{-2 \pi^2 t \nu} \pi\, \mathrm{Sin}[\pi\,(-t + x)]\, \mathrm{Sin}[\pi\,(-t + y)] \end{pmatrix}$$

```
(* Laplacian of the velocity field *)
```

```
lapu =
  FullSimplify[{D[u[[1]], x, x] + D[u[[1]], y, y], D[u[[2]], x, x] + D[u[[2]], y, y]}];
MatrixForm[
 lapu]
```

$$\begin{pmatrix} 2\,A\,e^{-2\pi^2 t\,\nu}\,\pi^2\,\text{Cos}[\pi\,(-t+x)]\,\text{Sin}[\pi\,(-t+y)] \\ 2\,A\,e^{-2\pi^2 t\,\nu}\,\pi^2\,\text{Cos}[\pi\,(t-y)]\,\text{Sin}[\pi\,(t-x)] \end{pmatrix}$$

```
(* Some forcing function that is set to 0 *)

g = {0, 0}
```

{0, 0}

```
(* Proceeding to construct a pressure field that works *)

rhs = Simplify[- (dudt + gradu.u - ν lapu - g)];
MatrixForm[rhs]
```

$$\begin{pmatrix} \frac{1}{2}\,A^2\,e^{-4\pi^2 t\,\nu}\,\pi\,\text{Sin}[2\,\pi\,(-t+x)] \\ -\frac{1}{2}\,A^2\,e^{-4\pi^2 t\,\nu}\,\pi\,\text{Sin}[2\,\pi\,(t-y)] \end{pmatrix}$$

```
int1 = Simplify[Integrate[rhs[[1]] , x]]
```

$$-\frac{1}{4}\,A^2\,e^{-4\pi^2 t\,\nu}\,\text{Cos}[2\,\pi\,(t-x)]$$

```
int2 = Simplify[Integrate[rhs[[2]] , y]]
```

$$-\frac{1}{4}\,A^2\,e^{-4\pi^2 t\,\nu}\,\text{Cos}[2\,\pi\,(t-y)]$$

```
p = FullSimplify[int1 + int2]
```

$$-\frac{1}{4}\,A^2\,e^{-4\pi^2 t\,\nu}\,(\text{Cos}[2\,\pi\,(t-x)]+\text{Cos}[2\,\pi\,(t-y)])$$

```
(* Does this pressure field work? *)

gradp = {D[p, x], D[p, y]}
```

$$\left\{-\frac{1}{2}\,A^2\,e^{-4\pi^2 t\,\nu}\,\pi\,\text{Sin}[2\,\pi\,(t-x)],\ -\frac{1}{2}\,A^2\,e^{-4\pi^2 t\,\nu}\,\pi\,\text{Sin}[2\,\pi\,(t-y)]\right\}$$

```
Simplify[gradp[[1]] - rhs[[1]]]
```

0

```
Simplify[gradp[[2]] - rhs[[2]]]
```

0

```
(* Looks good. Checking everything in the Navier-Stokes equation *)

MatrixForm[Simplify[dudt + gradu.u + gradp - ν lapu - g]]
```

$$\begin{pmatrix} 0 \\ 0 \end{pmatrix}$$

```
(* Good, we have our anlytical solution. Initial conditions
 and boundary conditions can also be suitably set from these. *)

CForm[u /. {ν → nu}]
```

```
List(1 - (A*Cos(Pi*(-t + x))*Sin(Pi*(-t + y)))/Power(E,2*nu*Power(Pi,2)*t),
   1 + (A*Cos(Pi*(-t + y))*Sin(Pi*(-t + x)))/Power(E,2*nu*Power(Pi,2)*t))
```

```
CForm[p /. {ν → nu}]
```

```
-(Power(A,2)*(Cos(2*Pi*(t - x)) + Cos(2*Pi*(t - y))))/
   (4.*Power(E,4*nu*Power(Pi,2)*t))
```

```
Plot3D[p /. {A → 1, ν → 1 / 8, t → 0}, {x, -1, 1}, {y, -1, 1}]
```